11. gyakorlat

(taszkok, kölcsönös kizárás, szemafor)

Példa 1

Mit csinál a következő program? Adjuk meg a feladat pontos kiírását.

with Ada.Text_IO; use Ada.Text_IO;

procedure Kocsma is

    type Ital is (Sör, Bor, Pálinka);

-----------

    task Kocsmáros is

        entry Tölt( Mit: in Ital );

    end Kocsmáros;

    task body Kocsmáros is

    begin

        for I in 1..20 loop

            accept Tölt ( Mit: in Ital ) do

                Put_Line("Töltök " & Ital'Image(Mit) & 't');

                case Mit is

                    when Sör => delay 1.0;

                    when Bor => delay 0.2;

                    when Pálinka => delay 0.3;

                end case;

            end Tölt;

        end loop;

    end Kocsmáros;

---------------------

    task type Részeg;

    task body Részeg(név:integer) is

        Sörivás_Ideje: Duration := 1.0;

    begin

        Kocsmáros.Tölt(Pálinka);

        Put_Line("Benyomok egy felest.");

        delay 0.1;

        Kocsmáros.Tölt(Bor);

        Put_Line("Benyomok egy pohár bort.");

        delay 0.3;

        loop

            Kocsmáros.Tölt(Sör);

            Put_Line("Benyomok egy korsó sört.");

            delay Sörivás_Ideje;

            Sörivás_Ideje := 2 * Sörivás_Ideje;

        end loop;

    end Részeg;

--------------------------

    type Részeg_Access is access Részeg;

    R: Részeg_Access;

begin

    for I in 1..5 loop

        delay 3.0;

        Put_Line("Egy részeg tévedt erre.");

        R := new Részeg(i);

    end loop;

end Kocsma;

Példa 2.


A feladat a következő: Imi 5-öst kapott Adából és ezt el akarja mondani telefonon egyik ismerősének. Az elsővel kezdi, fehívja és vár egy kicsit, hogy ő felvegye. Ha nem veszi fel az első ismerőse, akkor hívja a másodikat és így tovább. (Most 3 ismerőse van.) Az ismerősök pedig úgy működnek, hogy bemennek a szobájukba, ha épp csörög a telefon felveszik, ha nem, akkor egy időre elhagyják a szobát. (Ilyenkor nem hallják a csörgést.) Szimuláljuk ezt az esetet addig amíg Iminek sikerül valakivel beszélnie.

 

 with Ada.Text_Io, Ada.Numerics.Discrete_Random;

use Ada.Text_Io;

procedure Telefon is 

   subtype Int is Integer range 1..3;

   package Irand is new Ada.Numerics.Discrete_Random(Int);

   use Irand;

---------------------

   type Ismeros_Tipus is 

         (Kati, 

          Jani, 

          Sanyi); 

---------------------

   task type Ismeros(Ism : Ismeros_Tipus) is -- diszkriminánsos!

      entry Hiv; 

   end Ismeros;

--------------------

   task Imi;

-------------------

   --A Sim taszk mondja   meg,  hogy tart - e még a szimuláció. 

   task Sim is

      entry Vege (

            B :    out Boolean ); 

      entry Legyen_Vege; 

   end Sim;

   task body Sim is

      Bb : Boolean := False;  

   begin

      loop

         select -- arra ugrik a vezérlés, ahol hívás van

            accept Vege (

                  B :    out Boolean ) do 

               B := Bb;

            end Vege;

         or

            accept Legyen_Vege do 

               Bb := True;

            end Legyen_Vege;

         or

            terminate;

         end select;

      end loop;

   end Sim;

-------------------

   task body Ismeros is  -- Ism a diszkriminánsa ennek a taszknak
      B : Boolean;  

      G : Generator;  

   begin

      Reset(G);

      delay Duration(Random(G)); -- az első ne legyen mindjárt bent

      Sim.Vege(B);

      while not B loop

         select

            accept Hiv do 

               Put_Line(Ismeros_Tipus'Image(Ism) &

                  ": sikerult velem beszelni");

               Sim.Legyen_Vege;

            end Hiv;

         else

            Put_Line(Ismeros_Tipus'Image(Ism) & ": nem volt dumcsi");

            delay Duration(Random(G));

         end select;

         Sim.Vege(B);

      end loop;

   end Ismeros;

-----------------------

   I1 : Ismeros (Kati);  

   I2 : Ismeros (Jani);  

   I3 : Ismeros (Sanyi);  

---------------------

   task body Imi is

      B : Boolean := False;  

   begin

      while not B loop

         select

            I1.Hiv;

            exit;

         or

            delay 1.0; --eddig vár a hívásának fogadására, utána mást csinál

            Put_Line("Nem sikerult Katival beszelnem");

         end select;

         select

            I2.Hiv;

            exit;

         or

            delay 1.0;

            Put_Line("Nem sikerult Janival beszelnem");

         end select;

         select

            I3.Hiv;

            exit;

         or

            delay 1.0;

            Put_Line("Nem sikerult Sanyival beszelnem");

         end select;

      end loop;

   end Imi;

begin

   null;

end Telefon; 

Példa 3. szem.adb -- szemafor

Hozzunk létre egy Kiir nevű taszk típust egy stringgel paraméterezhető belépési ponttal. A Kiir tipusú taszkok a paraméterben kapott stringről készítsenek egy stringre mutató pointerbe egy másolatot és 1-től a Positive értéktartomány végéig írják ki a kapott sztringet magával a pozitív számmal együtt.  Futassunk két taszkot párhuzamosan, melyek a "jancsi" ill. "juliska" sztringgel hívják meg a Kiir taszkot. 

Egyszerű szemafor segítségével biztosítsunk kölcsönös kizárást a számokat ill. a Jancsi és Juliska sztringet kiíró program részre. 

Szemaforral: 

with Text_IO, Ada.Command_Line;

use Text_IO, Ada.Command_Line;

procedure Jancsi_Julcsa5 is

   type String_Mut is access String;

   task type Kiiro ( Nev : String_Mut := new String' ( "" ));


   task Szemafor is

      entry P; 

      entry V; 

   end Szemafor;


   task body Kiiro is


      procedure Kiir ( S : in String ) is

      begin

         for I in S'range loop

            Put ( S ( I ));

         end loop;

         New_Line;

      end Kiir;

   begin

      for I in Positive'range loop

         Szemafor.P;

         Kiir ( Nev.all & ": " & Positive'Image ( I ));

         Szemafor.V;

         if Argument_Count > 0 then

            delay Duration'Value ( Argument ( 1 ));

         end if;

      end loop;

   end Kiiro;


   task body Szemafor is

   begin

      loop

         accept P;

         accept V;

      end loop;

   end Szemafor;

   Jancsi : Kiiro ( new String' ( "Jancsi" ));

   Julcsa : Kiiro ( new String' ( "Julcsa" ));

begin

   null;

end Jancsi_Julcsa5;

Példa 4. 

Mit csinál a "pelda4" alkönyvtárban megadott program? 

